関数のアクアリウム (NO.1)
逐次代入法・螺旋

SHIMURA Masato
JCD02773@nifty.ne.jp

2010 年 12 月 6 日

目次

はじめに

関数を色々な魚のように飼って観察し、大いに楽しもう。

ダンハムは解析学の幾何学的直感からの決別はオイラーから始まり、ワイエルシュトラウスが棺に最後の釘を打ったと記する。非線形、複雑系やカオスなど分野や、経済分野などでは図表やグラフは必須である。

水族館と異なるのは、J をダウンロードして短いスクリプトを打込めば自宅の水槽で飼育可能なことである。

準備 require 'plot trig numeric'

1 逐次代入法の水槽

逐次近似法は [高速・尺取虫] で、単純明快、最近のマシーンでは十分に早い。

数式の解（X 軸と交叉する点）ではなく関数全体の振る舞いをとらえる。不動点や周期点、吸引点が重要となる。

学校数学では漸加関係は収束、発散（沈む、浮く）に注力するようだが、ここではグラフィックスを多用して魚の姿を鑑賞しよう

1.1 逐次代入法による漸加式

カオスや蜘蛛の巣の軌跡を辿りたいときは下手にアクセラレーションをかけないシンプルな逐次代入法が都合良い。J は漸化式が簡単に計算できる。

$$x_{n+1} = F(x_n)$$

■Example

$$x_{n+1} = 0.5x_n + 4$$

$$\begin{array}{lllll}
x_1 = & 0.5\times & 0 & +4 = & 4 \\
x_2 = & 0.5\times & 4 & +4 = & 6 \\
x_3 = & 0.5\times & 6 & +4 = & 7 \\
x_3 = & 0.5\times & 7 & +4 = & 7.5 \\
\cdots & & & &
\end{array}$$

`4 0.5&p. ^:(i.30) 4 NB. 4` **は初期値**

```
4 7 $  4 0.5&p. ^:(i.28) 4
      4       6       7      7.5     7.75    7.875   7.9375
7.96875 7.98438 7.99219 7.99609 7.99805 7.99902 7.99951
7.99976 7.99988 7.99994 7.99997 7.99998 7.99999       8
      8       8       8       8       8       8       8
```



p.　多項式であるので J の多項式のプリミティブ (p.) を用いる。
反復　`^:(i.28)` は J の簡易反復で i.n とすると経過が得られる

上の例では $x_{n+1} = 0.5x_n + 4$ は初期値は 4 としたが、8 や 12 から出発しても極限は 8 になる。$8 = F(8)$ となるので 8 は**不動点**である。

横軸に $x_n = present$, 縦軸に $x_{n+1} = next$ を取ると $x_{n+1} = x_n$ は 45^o 線となり、グラフと 45 度線との交点は不動点となる。

不動点は次の連立方程式からも求められる。

$$\begin{cases} x_{n+1} = & x_n \\ x_{n+1} = & 0.5x_n \quad +4 \end{cases}$$

1.1.1 **関数計算の補助** Script

- 反復計算

```
 calc_com=: 1 : 0
NB. Usage:  4 0.5&p.  calc_com 21 0
'repeat fval'=: y  NB. repeat times & first value
u ^:(i. repeat) fval
)
```

- グラフィックス用の補助関数
  - グラフィックス用の $x_{n+1} = x_n$ の生成
    一期前の結果 ($x_n = present$) に関数を作用させたとき ($x_{n+1} = next$) に、present を X 軸に next を Y 軸に取るように配置する
    ```
     com_form=: 3 : ' { |:  ;("1),. 2<\ y'
    NB.Usage: 'line,marker' plot com_form 4 _0.5&p. ^:(i.21) 0
    ```
  - タートルグラフィックス
    蜘蛛の巣を生成する

1.2 cos

$cos(1)$ の逐次代入

$cos(cos(cos(1)))$

```
'marker,line' plot  2&o. ^:(i.50) 1
```

```
   2&o. ^:(_) 1
0.739085
```



1.3 ロジスチック差分方程式

ロジスチック写像

$$x_{n+1} = bx_n(1 - x_n)$$

ロジスティック写像は周期倍分岐が見られ、不動点から 2 周期軌道、4 周期軌道、8 周期軌道を経てカオスに至る。

発見 1844,5 年 *Pierre Francois Verhulst(1804-1849)*

人口問題のモデルとして提唱された

同一種, 亜種

$$\frac{dx}{dt} = (a - bx)x \quad (a > 0, b > 0)$$

$$f(t) = \frac{1}{1 + ce^{-at}}$$



ロジスティック写像から生成した時系列



カオスの図形解析（X 軸=x_n,Y 軸=x_{n+1}）

```
plot  3.8&fL0 ^:(i.120) 0.4
'marker' plot  com_form 3.8&fL0 ^:(i.120) 0.4
```

Script `fL0=: 4 : '(x* y)* (-. y)'`

```
NB. Usage: plot  com_form 3.8&fL0 calc_com 200 0.4
```

1.3.1 熊手のロジスティクス

Logistic and Caos

`lg=: * * -. @ ]` NB. C.Reiter



```
'color blue' plot (steps 1 4 100)&   lg ^:(100+i.20) 0.1
   pd 'eps /temp/aqua_logis3.eps'
```

次のようにしても

```
plot  (steps 1 4 100)&fL0 ^:(i.100) 0.1
```

1.3.2　テント写像

テント写像

$$x_{n+1} = 1 - |1 - 1.9x_n|$$

計算値とグラフ　.





ロジスティック写像から生成した時系列　　　　カオスの図形解析（X 軸=x_n,Y 軸=x_{n+1}）

Script　`fT0=: 4 : '-. | -. x * y' NB. x(n+1)=1-|1-1.9x(n)|`

`NB. Usage: 'marker,line' plot com_form 1.9&fT0 calc_com 50 0.1`

生態　カントール 3 進集合の類縁

1.4　線形連立方程式の漸化式

1.4.1　フィボナッチ数

フィボナッチ数

数式（マトリックスフォーム）

$$\begin{bmatrix} F_{k+1} \\ F_{k+2} \end{bmatrix} = \begin{bmatrix} 0 & 1 \\ 1 & 1 \end{bmatrix} \begin{bmatrix} F_k \\ F_{k+1} \end{bmatrix}$$

数式

$$F_n = \begin{cases} 0 & if\ n = 0 \\ 1 & if\ n = 1 \\ F_{n-1} + F_{n-2} & if\ n > 1 \end{cases}$$

発見　フィボナッチの本当の名はレオナルド・ダ・ピサ。ピサ出身のレオナルド（1170 頃-1250 頃)。父が現在のアルジェリアに仕事を求めたのに従いアラブ世界で最新の数学を学ぶ。1202 年に帰国し「算盤の書」を表し、アラビア数学を後進国ヨーロッパに広めた。

一つがいの兎は生まれて *2* ヶ月後から毎月一つがいの兎を生む

一つがいの兎は *1* 年の間に何つがいの兎になるか

Script　`(0 1 ,:1 1)&(+/ . *) ^:(i.10) 1 1`

遊泳エンジン　マトリクス計算のため外積 *(+/ . *)* を外付け

.

```
a=. (0 1 ,:1 1)&(+/ . *) ^:(i.16) 1 1

  a,. %/("1) |."1 a
  1    1        1
  1    2        2
  2    3      1.5
  3    5  1.66667
  5    8      1.6
  8   13    1.625
 13   21  1.61538
 21   34  1.61905
 34   55  1.61765
 55   89  1.61818
 89  144  1.61798
144  233  1.61806 NB. ans is 233 and 144 is 89+55
233  377  1.61803 NB.（当月生れ 89 1 月経
過 55 2月以上 89 計233)
377  610  1.61804
610  987  1.61803
987 1597  1.61803
```



黄金比　$k \to \infty$，のとき $\dfrac{f_{k+1}}{f_k}$ は黄金比 $1 : \dfrac{1+\sqrt{5}}{2}$ に近づく。

生息域　オーム貝やひまわりの黄金螺旋

連分数　フィボナッチ数列 F_n の各項の比から作られる数列

$$r_n = \frac{F_{n-a}}{F_n}$$

は次の漸化式を満足させる

$$r_1 = 0, \quad r_{n+1} = \frac{1}{r_n + 1}$$

$$r_6 = \cfrac{1}{1 + \cfrac{1}{1 + \cfrac{1}{1 + \cfrac{1}{1 + 1}}}}$$

```
   (+%)/\ 10$1x
1 2 3r2 5r3 8r5 13r8 21r13 34r21 55r34 89r55
```

黄金比（2）　.

```
 4 5 $ _1 x:  (+%)/\ 20$1x
```

```
      1       2     1.5 1.66667     1.6
  1.625 1.61538 1.61905 1.61765 1.61818
1.61798 1.61806 1.61803 1.61804 1.61803
1.61803 1.61803 1.61803 1.61803 1.61803
```

1.5　蜘蛛の巣

蜘蛛の巣

$$x_{n+1} = -0.5x_n + 4$$

右肩上がりの供給曲線と右肩下がりの需要曲線がおりなす X に巻き付く糸は蜘蛛の巣曲線として知られ、経済学の教科書の定番であった。

2　螺旋の水槽

2.1　アルキメデス螺旋

Archimedes Spiral

$$r = a\theta$$

$$\varphi(\theta) = \begin{pmatrix} r cos\theta \\ r sin\theta \end{pmatrix}$$



```
1 archi steps 0 24p1 1000
```

Script

```
archi=: 4 : 0
NB. Archimedes spiral
NB. x is a //ex 0.1
NB.  1 archi steps 0 24p1 1000
r=: x * y
x0=: r * 2&o. y
y0=: r * 1&o. y
```

```
plot x0;y0
)
```

生息域　. SP/LP レコード

ヤコブ · ベルヌイの墓石（間違ってアルキメデス螺旋を彫られた）

2.2　ベルヌイ螺旋

ベルヌイ螺旋 · 対数螺旋 · 等角螺旋

$$r = ae^{b\theta}$$

$$\varphi(\theta) = \begin{pmatrix} ae^{b\theta}cos\theta \\ ae^{b\theta}sin\theta \end{pmatrix}$$



`1 1r100p1 ber steps 0 24p1 1000`

Jacob·Bernoulli(1654-1705 **スイス**)

Script　`bern=: 4 : 0`

```
NB. Bernoulli spiral
NB.  1 1r100p1 bern steps 0 24p1 1000
'a b'=. x
r=:  a* 1x1 ^ b*y
x0=: r * 2&o. y
y0=: r * 1&o. y
plot x0;y0
)
```

生息域　オーム貝など貝類 (黄金螺旋参照)

パラメータを変えて楽しんでください

`1 1r10p1 bern steps 0 24p1 10000`



2.2.1 J Grammar

1p1 π

1r3p2 $\frac{1}{3}\pi^2$

2.3 円の伸開線

円形の糸巻きを解くときの糸の先端の描く曲線

Evolute spiral

$$x = a(cos\theta + \theta sin\theta)$$

$$y = a(sin\theta + \theta cos\theta)$$



```
evolute steps 0 2p1 1000
```

Script

```
evolute=: 3 : 0
NB. evolute spiral
NB.   evolute steps 0 2p1 1000
t=: 2p1* y
a=. 1 NB. radius of circle
x0=: a * (2&o. y) + t * sin t
y0=: a * (1&o. y)+ t * cos t
```

```
plot x0;y0
)
```

2.4 3D スパイラル

J の LAB の math に 3DSpiral が入っていた

定義（1） `dfr=: *&(180%1p1)   NB.*dfr v degrees from radians`

```
rfd=: *&(1p1%180)   NB.*rfd v radians from degrees
```

3Dcircle(1) `spiral=: cos&rfd ; sin&rfd ; %:`

```
plot spiral i. 10000
```



3Dcircle(2) `decay=: ^@(%&_500)`

```
desin=: decay * sin&rfdd
decos=: decay * cos&rfd
despi=: decos ; desin ; %:
plot despi i. 1000
```



References

有賀祐二「進化経済学の数理入門」共立出版　2004
西沢 · 関口 · 吉野　「フラクタルと数の世界」海文堂　1991
松葉育雄「複雑型の数理」朝倉書店　2004

Miscellance

J(602) is DL available
`http://www.jsoftware.com`
Platform WIN32/64 LINUX32/64 MAC/Intel/PPC PocketPC
Script is DL from
`http://japla.sakura.ne.jp`

スクリプトの扱い方

J（本体）は DL してデフォルトでインストールすればよい。J はプロパティを利用していないのでディレクトリ（フォルダ）を選ばない。COPY すれば USB や CDROM で持ち運びでき、どの PC でも（インストールしないで）利用できる。

スクリプトは IJS に書く。読み込んだものも IJS に入るる。
これをアクティブにするには ijs の画面で Ctrl+W が一番簡単
Run File Run Windou をマウスやキーボードでも可能
ijx の画面で実行する　F12 や Alt+Tab でも移れる